Example 4: a system of ODEs in different intervals

Consider a two-phase flow[11]

(i) Region 1

$$\frac{d^2u_1}{dy^2} + \frac{Gr}{Re}\sin(\phi)\theta_1 = P, \tag{22}$$

$$\frac{d^2\theta_1}{dy^2} + PrEc\left(\frac{du_1}{dy}\right)^2 = 0, \tag{23}$$

(ii) Region 2

$$\frac{d^2u_2}{dy^2} + \frac{Gr}{Re}\frac{nbh^2}{\lambda}\sin(\phi)\theta_2 - \frac{M^2h^2}{\lambda}u_2 = \frac{h^2}{\lambda}P, \tag{24}$$

$$\frac{d^2\theta_2}{dy^2} + EcPr\frac{\lambda}{\lambda_T}\left(\frac{du_2}{dy}\right)^2 + EcPr\frac{h^2}{\lambda_T}M^2u_2^2 = 0, \tag{25}$$

subject to

$$u_1(1) = 1,\ \theta_1(1) = 1, \tag{26}$$

$$u_1(0) = u_2(0),\ \theta_1(0) = \theta_2(0), \tag{27}$$

$$u_1'(0) = \frac{\lambda}{h}u_2'(0),\ \theta_1'(0) = \frac{\lambda_T}{h}\theta_2'(0), \tag{28}$$

$$u_2(-1) = 0,\ \theta_2(-1) = 0, \tag{29}$$

where Gr is the Grashof number, Ec is the Eckert number, Pr is the Prandtl number, Re is the Reynolds number, M is the Hartmann number and P is the dimensionless pressure gradient. This model describes a two-fluid magnetohydrodynamic Poiseuille-Couette flow and heat transfer in an inclined channel. Umavathi[11] investigate this model analytically by regular perturbation method and numerically by finite difference technique.

The BVPh 2.0 can solve this problem (22)–(29) directly without difficulty. Since all the parameters in the system will be given, we have `NumEQ = 4` and `TypeEQ=1`. The system is input as

```
TypeEQ = 1;
NumEQ = 4;
f[1,z_,{u1_,s1_,u2_,s2_},Lambda_]:=
    D[u1, {z, 2}]+ Gr/Ra*Sin[phi]*s1 - P ;
f[2,z_,{u1_,s1_,u2_,s2_},Lambda_]:=
    D[s1, {z, 2}] + Pr*Ec*(D[s1, z])^2;
f[3,z_,{u1_,s1_,u2_,s2_},Lambda_]:=D[u2,{z, 2}]-h^2/lamb*P
    +Gr/Ra*Sin[phi]*n*b*h^2/lamb*s2-M^2*h^2/lamb*u2;
f[4,z_,{u1_,s1_,u2_,s2_},lambda_]:=D[s2,{z,2}]
    +Pr*Ec*lamb/lambT*D[u2,z]^2+Pr*Ec*h^2/lambT*M^2*u2^2;
```

The eight boundary conditions (26)–(29) are defined as

```
NumBC=8;
BC[1,z_,{u1_,s1_,u2_,s2_}]:=(u1-1)/.z->1;
BC[2,z_,{u1_,s1_,u2_,s2_}]:=(u1-u2)/.z->0;
BC[3,z_,{u1_,s1_,u2_,s2_}]:=u2/.z->-1;
BC[4,z_,{u1_,s1_,u2_,s2_}]:=(D[u1,z]-D[u2,z]*lamb/h)/.z->0;
BC[5,z_,{u1_,s1_,u2_,s2_}]:=(s1-1)/.z->1;
BC[6,z_,{u1_,s1_,u2_,s2_}]:=(s1-s2)/.z->0;
BC[7,z_,{u1_,s1_,u2_,s2_}]:=s2/.z->-1;
BC[8,z_,{u1_,s1_,u2_,s2_}]:=(D[s1,z]-D[s2,z]*lambT/h)/.z->0;
```

Now let us input the solution intervals

```
    zL[1] = 0;  zR[1] = 1; (* u1 *)
    zL[2] = 0;  zR[2] = 1; (* s1 *)
    zL[3] = -1; zR[3] = 0; (* u2 *)
    zL[4] = -1; zR[4] = 0; (* s2 *)
```

Note that the solution intervals are not the same. Since all the solution intervals are in finite intervals, we do not have to specify the integral interval to compute the squared residual error.

The initial guesses are chosen as $u_{1,0} = \frac{\lambda}{h}(z-z^2)+1$, $\theta_{1,0} = \frac{z\lambda_T}{h}+(1-\frac{\lambda_T}{h})z^2$, $u_{2,0} = 1+z$ and $\theta_{2,0} = z+z^2$. They are input as

```
    U[1, 0] = (z - z^2)*lamb/h+1;           (* u1 *)
    U[2, 0] = z*lambT/h +(1-lambT/h)*z^2;   (* s1 *)
    U[3, 0] = 1 + z;                        (* u2 *)
    U[4, 0] = z^2 + z;                      (* s2 *)
```

The auxiliary linear operators are chosen as $\mathcal{L}_1 = \mathcal{L}_2 = \mathcal{L}_3 = \mathcal{L}_4 = \frac{\partial^2}{\partial y^2}$. They are defined as

```
    L[1,u_]:=D[u,{z,2}];
    L[2,u_]:=D[u,{z,2}];
    L[3,u_]:=D[u,{z,2}];
    L[4,u_]:=D[u,{z,2}];
```

Note that we use the delayed assignment `SetDelayed(:=)` to define these linear operators and z is the independent variable in the package.

Without loss of generality, let us consider the case when $Pr = 7/10$, $Ec = 1/100$, $P = -5$, $b = 1$, $n = 1$, $Re = 1$, $M = 2$, $Gr = 5$, $h = 1$, $\lambda = 1$, $\lambda_T = 1$, and $\phi = \pi/6$. These physical parameters are input as

```
P = -5;  b = 1;
n = 1;   Ra = 1;
M = 2;   Gr = 5;
lamb = 1;  lambT = 1;
h = 1;  phi = Pi/6;
Pr = 7/10; Ec = 1/100;
```

At this time, we have input all the data for this problem, except the convergence-control parameters `c0[k]`. We minimize the squared residual error of the 4th-order approximations to obtain optimal values for `c0[k]` by the command

```
GetOptiVar[4,{},{c0[1],c0[2],c0[3],c0[4]}];
```

Note that the second parameter of `GetOptiVar` is a empty list, which means that we give no constraints on the convergence-control parameters `c0[k]`.

After some time , we obtain the optimal values for `c0[k]`, which reads `c0[1]` ≈ -0.898166, `c0[2]` ≈ -0.946828, `c0[3]` ≈ -0.780946 and `c0[4]` ≈ -1.12363. Then we call the main module `BVPh` to get the 30th-order approximations

```
BVPh[1, 30];
```

The 30th-order approximations for u_1, θ_1, u_2, θ_2 are stored in `U[1,30]`, `U[2,30]`, `U[3,30]` and `U[4,30]`, respectively, while the corresponding squared residual error is `ErrTotal[30]`. The 30th-order approximations are plotted in Fig. 10. The value of $\theta(y)$ agrees with Umavathi's result[11] (black dots), as shown in Fig. 10. The 30th-order approximation of $\theta(y)$ gives the heat transfer rate $Nu_+ = \theta_1'(1) = 0.8860625$ and $Nu_- = \theta_2'(1) = 1.122312$, which agrees with $Nu_+ = 0.88606$ and $Nu_- = 1.12230$ in Umavathi's[11] Table 3.

The total error of the system for every two order of approximations is plotted in Fig. 11.



Fig. 10: The curve of $u(y)$ (solid) and $\theta(y)$ (dashed) for Example 4. The black dots are the values for $\theta(y)$ obtained by Umavathi.[11]



Fig. 11: Total error vs. order of approximation for Example 4.

References

1. S. J. Liao, *Proposed homotopy analysis techniques for the solution of nonlinear problem.* Ph.D. thesis, Shanghai Jiao Tong University (1992).

2. S. J. Liao, A uniformly valid analytic solution of two-dimensional viscous flow over a semi-infinite flat plate, *J. Fluid Mech..* **385**, 101–128 (1999).

3. S. J. Liao, On the analytic solution of magnetohydrodynamic flows of non-newtonian fluids over a stretching sheet, *J. Fluid Mech..* **488**, 189–212 (2003).

4. S. J. Liao, Series solutions of unsteady boundary-layer flows over a stretching flat plate, *Stud. Appl. Math..* **117**(3), 239–263 (2006).

5. S. J. Liao, *Beyond Perturbation—Introductioin to the Homotopy Analysis Method.* Chapman & Hall/CRC Press, Boca Raton (2003).

6. S. J. Liao, *Homotopy Analysis Method in Nonlinear differential equations.* Springer-Verlag Press, New York (2011).

7. S. J. Liao, Notes on the homotopy analysis method: Some definitions and theorems, *Commun. Nonlinear Sci. Numer. Simulat..* **14**, 983–997 (2009).

8. M. Sajid, Z. Iqbal, T. Hayat and S. Obaidat, Series solution for rotating flow of an upper convected Maxwell fluid over a strtching sheet, *Commun. Theor. Phys..* **56**(4), 740–744 (2011).

9. T. Hayat, M. Nawa and A. A. Hendi, Heat transfer analysis on axisymmetric MHD flow of a micropolar fluid between the radially stretching sheets, *J. Mech..* **27**(4), 607–617 (2011).

10. H. Xu, T. Fan and I. Pop, Analysis of mixed convection flow of a nanofluid in a vertical channel with the Buongiorno mathematical model, *Int. Commun. Heat Mass.* **44**, 15–22 (2013).

11. J. C. Umavathi, I. C. Liu and J. Prathap Kumar. Magnetohydrodynamic Poiseuille-Couette flow and heat transfer in an inclined channel, *J. Mech..* **26**(4), 525–532 (2010).

12. J. P. Boyd, An analytical and numerical study of the two-dimensional Bratu equation, *J. Sci. Comput..* **1**(2), 183–206 (1986).

13. J. Jacobsen and K. Schmitt, The Liouville-Bratu-Gelfand problem for radial operators, *J. Differ. Equations.* **184**, 283–298 (2002).

14. J. S. McGough, Numerical continuation and the Gelfand problem, *Appl. Math. Comput..* **89**(1-3), 225–239 (1998).

15. Y. L. Zhao, Z. L. Lin and S. J. Liao, An iterative analytical approach for nonlinear boundary value problems in a semi-infinite domain, *Comput. Phys. Commun..* Online.